
Hoofdstuk 7.

Procedures.

In hoofdstuk 1 werd het belang aangetoond van gestructureerde programma's bestaande uit een hierarchie van geneste deelprogramma's, elk met een wel gedefinieerde taak. PROCEDUREs en MODULEs laten toe deze strategie te implementeren in Modula 2.

7.1. Het concept subroutine.

Een van de ideeën achter procedures, het concept subroutine, wordt geïllustreerd in fig. 7.1. In dit voorbeeld bevindt zich een bepaald stuk code CCC op twee verschillende plaatsen. In plaats van deze code twee maal te voorzien, kan men deze code laten deel uitmaken van de subroutine DoCCC en, op het ogenblik dat men CCC nodig heeft, wordt DoCCC aangeroepen. De uitvoering van de DoCCC instructie veroorzaakt een sprong naar de eerste instructie van de DoCCC subroutine. Wanneer de CCC code volledig is uitgevoerd, bv. wanneer een RETURN instructie of gewoonweg het einde van de subroutine bereikt wordt, wordt de uitvoering verdergezet vanaf de instructie die de sprong naar DoCCC opvolgt.

Fig.7.1. Subroutines

Aangezien subroutines vanop verschillende plaatsen in het programma kunnen aangeroepen worden, is het de taak van de stuur eenheid, in geval er naar de subroutine gesprongen wordt, het adres van de sprong instructie te onthouden om achteraf de uitvoering op de juiste plaats te kunnen verder zetten. Daarvoor schrijft de stuur eenheid dat adres in het gegevensgeheugen wanneer een subroutine opgeroepen wordt en haalt het het terug op bij de uitvoering van een RETURN instructie.

7.2. Procedures.

In Modula 2 worden subroutines geïmplementeerd als procedures. Dat zijn subroutines die een veel uitgebreidere functionaliteit hebben dan de hierboven besproken subroutines. Ze laten eveneens toe de draagwijdte van identifiers te beperken tot de programma delen waar deze identifiers werkelijk bereikbaar moeten zijn en zorgen voor het beheer van het gegevens geheugen van de computer tijdens de uitvoering van programma's. Elke procedure kan eigen (lokale) variabelen hebben en deze variabelen bestaan (dwz. verwijzen naar een fysische plaats in het gegevensgeheugen) enkel tijdens de uitvoering van de procedure, zodanig dat, wanneer een procedure niet aktief is, de plaats nodig voor de variabelen van die procedure herbruikt kan worden voor variabelen van andere procedures. 

7.2.1. Blok gestructureerde talen.

De syntax van modules en procedures zijn gelijkaardig, zoals aangetoond wordt in fig.7.2,  waarin de vereenvoudigde syntax diagrammen voor modules en procedures te zien zijn: ze hebben beiden een hoofding, bestaande uit de naam, gevolgd door een beschrijving van de uitwisselingen die kunnen gebeuren met de buitenwereld (IMPORT/EXPORT lijst en parameters), en uiteindelijk een BLOCK. Het concept van BLOCK is essentiëel voor de zogenaamde blok-gestructureerde talen zoals ALGOL, Pascal, Modula 2 en Ada.

 Fig.7.2. Simplified MODULE and PROCEDURE syntax diagrams.

Fig.7.3 toont de syntax van een Modula 2 blok. Merk op dat een blok in feite het belangrijkste deel van het programma is: het bevat de beschrijving van de gegevens en de uitvoerbare instructies, en, tussen beiden, de definitie van procedures (de zogenaamde in het blok geneste procedures), die allen zelf een blok bevatten. Een programma in een blok-gestructureerde taal kan dus beschouwd worden als een verzameling van geneste blokken, bestaande uit gegevens declaraties, uitvoerbare instructies en eventueel andere blokken. Deze structuur wordt schematisch voorgesteld in fig.7.4.

 Fig.7.3. Blok syntax.

 Fig.7.4. Geneste blokken.

7.2.2. Draagwijdte van identifiers (Scope of identifiers).
Aangezien identifiers in iedere blok kunnen gedefiniëerd worden, is het essentiëel te weten in welke blokken welke identifiers kunnen gebruikt worden. De hoofdregel hiervoor is uiterst eenvoudig: elke identifier kan overal binnen de blok waarin hij gedefinieerd is gebruikt worden, met inbegrip van de geneste blokken.

Een identifier wordt lokaal genoemd aan de blok waarin hij gedefinieerd is. Hij is globaal in alle blokken die genest zijn binnen de blok waarin hij gedefinieerd is. Fig.7.5 illustreert deze regel: de cardinal variabele A, gedefinieerd in de buitenste blok “BLUE” kan overal gebruikt worden; de real variabele B, gedefinieerd in de blok “PURPLE” kan overal binnen deze blok gebruikt worden, maar niet in het deel van de blok “BLUE” dat niet deel uitmaakt van de blok “PURPLE”; de character variabele C, gedefineerd binnen de blok “RED” kan uitsluitend binnen deze blok gebruikt worden.

Fig.7.5. Globale identifiers - data.

Fig.7.6. Lokale herdefinitie van globale identifiers.

Dezelfde identifier mag in twee verschillende declaraties in geneste blokken voorkomen. In dit geval heeft de lokale definitie voorrang op de globale, zoals geïllustreerd in fig.7.6, waar men ziet dat de identifier A die in de blok “BLUE” gedefinieerd werd als een CARDINAL variabele in de blok “RED” geherdefinieerd wordt als een constante. De identifier A stemt dus overeen met een CARDINAL variabele in gans het programma, behalve in de blok “RED” waar A een constante string is. 

7.2.3. Terug naar het GGD voorbeeld.

MODULE GCD;
(* Version with procedures and global variables *)

FROM InOut IMPORT WriteLn, WriteString, ReadCard, WriteCard;

VAR x,y:CARDINAL;

PROCEDURE AskNumbers;

BEGIN


WriteString('Enter x and y ');


ReadCard(x);


ReadCard(y);

END AskNumbers;

PROCEDURE ComputeGCD;

BEGIN


WHILE x # y DO



IF x > y 




THEN x := x-y




ELSE y := y-x



END;


END;

END ComputeGCD;

PROCEDURE WriteResult;

BEGIN


WriteLn;


WriteString('The GCD is ');


WriteCard(x,6);

END WriteResult;
BEGIN

(* Here starts execution of the program *)

AskNumbers;

ComputeGCD;

WriteResult
END GCD.

Tabel.7.1. Het GGD programma met procedures en globale variabelen.

Als we terugkijken naar het voorbeeld van de GGD, zou men gebruik kunnen maken van procedures om de drie programma delen duidelijk te scheiden. Drie procedures, AskNumbers, ComputeGCD en WriteResult zullen gebruikt worden. Het hoofdprogramma zal enkel bestaan uit het aanroepen van drie procedures, met nadruk op de opeenvolgende niveaus van abstractie in de voorgestelde oplossing. De variabelen x en y worden gedeclareerd in het hoofdprogramma en zijn bijgevolg globaal in de drie procedures. Tabel 7.1 toont de nieuwe versie van het programma.

7.2.4. Toegankelijkheid van procedures.

Aangezien procedures gedeclareerd worden in blokken, gelden alle hogervermelde regels ook voor namen van procedures, wat toelaat na te gaan van waaruit welke procedure aangeroepen kan worden. Fig.7.7 toont verschillende mogelijke situaties: in de buitenste blok “BLUE” kunnen de procedures “YELLOW” en “PURPLE”, die lokaal zijn, aangeroepen worden. Binnen “PURPLE” kunnen de procedures “RED” en “GREEN”, die beiden lokaal zijn, maar ook “YELLOW” en “PURPLE”, die globaal zijn, aangeroepen worden. Dezelfde lijst van procedures kan aangeroepen worden vanuit “GREEN”, aangezien alle identifiers globaal zijn.

Fig.7.7. Globale identifiers - procedures.

Vanuit de procedure “ORANGE” zouden, rekening houdend met de draagwijdte regels van identifiers, procedures “YELLOW”,”PURPLE” en “ORANGE” toegankelijk moeten zijn, maar desalniettemin, in sommige versies van Modula 2 kunnen zich problemen voordoen wanneer verwijzingen naar verderop worden gedaan (d.w.z. dat een identifier gebruikt wordt vooraleer hij gedeclareerd wordt in de programma tekst). Deze problemen zijn het gevolg van een zwakke implementatie van de taal en hebben niets te maken met de semantiek van de taal. In verschillende implementaties van de taal wordt deze moeilijkheid opgevangen door de FORWARD declaratie: wanneer men een procedure moet gebruiken alvorens ze gedeclareerd is moet men aan de compiler melden dat de declaratie later komt door de naam van de procedure gevolgd door het gereserveerde woord FORWARD te vermelden.

7.3. Geheugen beheer voor blok gestructureerde talen.

7.3.1. Activatie Records.

Bij het compileren van blok gestructureerde talen is het niet opportuun onmiddellijk met elke variabele een vast adres in het datageheugen te associëren. In plaats daarvan gaat de compiler te werk alsof er voor elke blok een appart datageheugen zou bestaan dat men het “activatie record” van de blok noemt. Elke lokale variabele wordt zodoende geassocieerd met een bepaalde plaats in een activatie record en het effectieve adres van de variabele tijdens de uitvoering van de procedure wordt bekomen door de optelling van het beginadres van het activatierecord en van de positie van de variabele in datzelfde record.

Bij het aanroepen van een procedure, wordt er plaats in het datageheugen gereserveerd voor het activatie record van die procedure en wanneer de uitvoering van de procedure beëindigd is, wordt die plaats terug vrij gegeven. Ter illustratie toont friguur 7.8 de opeenvolgende inhouden van het gegevensgeheugen tijdens de uitvoering van het programma dat links in de figuur voorgesteld is. Ondanks in fig.7.8. gemakkelijkheidshalve alle activatie records met dezelfde afmetingen getoond worden is het vanzelfsprekend dat de omvang van een activatie record verschillend kan zijn voor elke procedure.

Fig.7.8. Opeenvolgende beelden van het datageheugen.

7.3.2. Statische en dynamische relaties.

Bij het oproepen van een procedure worden de eerstvolgende vrije plaatsen van het gegevensgeheugen gereserveerd voor het terugkeeradres en voor het activatie record van die procedure. Teneinde de adressen van de lokale variabelen efficiënt te kunnen berekenen, wordt het begin adres van het activatie record in een ad hoc geheugen in de stuureenheid, het “basisregister”, bewaard. Wanneer een procedure afgewerkt is, moet de inhoud van het basisregister terug gezet worden naar het begin van het activatie record van de blok waaruit de procedure opgeroepen was. Om dit mogelijk te maken wordt systematisch, wanneer een procedure opgeroepen wordt, de oude inhoud van het basisregister neergeschreven in een daarvoor gereserveerde plaats in het activatie record van de nieuwe procedure. Bij het tergugkeren volstaat het dan deze opgeslagen waarde terug in het basisregister te schrijven waardoor automatisch, het activatie record van de afgewerkte procedure onbereikbaar wordt.

Indien men de adressen, die bewaard worden in het basisregister of in een activatie record, schematisch voorstelt door middel van een pijl naar het aangeduide adres, kan men het gegevensgeheugen tijdens de uitvoering van een programma voorstellen zoals in figuur 7.9. Deze figuur stemt overeen met momentopname 3 van figuur 7.8. De verzameling van de adressen, die bewaard worden om bij het terugkeren uit een procedure de vorige aktivatieblok te kunnen terugvinden, noemt men de “dynamische banden”.

Fig,7.9. Dynamische banden in het gegevensgeheugen.

Toegang tot lokale variabelen is eenvoudig dankzij het hierboven beschreven mechanisme dat gebruik maakt van het basisregister. Toegang tot globale variabelen is echter complexer. Uit het voorbeeld van fig.7.8 zou men kunnen afleiden dat de globale variabelen gewoonweg te vinden zijn door de dynamische banden te volgen. Een aandachtige analyze van fig.7.10 toont aan dat niets minder waar is. 

[image: image10.wmf]BEGIN 

 PURPLE 

END 

BLUE

.

MODULE BLUE

BEGIN 

 GREEN;...

 END 

PURPLE

;

BEGIN … END 

RED;

BEGIN  YELLOW; END 

GREEN

;

PROCEDURE PURPLE;

PROCEDURE RED;

PROCEDURE GREEN;

BEGIN …;

 ORANGE; 

END 

YELLOW

;

BEGIN …;

...

 END 

ORANGE

;

PROCEDURE YELLOW;

PROCEDURE ORANGE;

BLUE

PURPLE

BLUE

GREEN

PURPLE

BLUE

1

2

3

4

5

9

6

7

8

10

BLUE

PURPLE

BLUE

PURPLE

BLUE

GREEN

PURPLE

BLUE

GREEN

YELLOW

PURPLE

BLUE

GREEN

YELLOW

PURPLE

BLUE

GREEN

YELLOW

ORANGE

PURPLE

BLUE

GREEN

YELLOW

Fig.7.10. Opeenvolgende beelden van het gegevensgeheugen. 

Hier wordt de procedure YELLOW opgeroepen vanuit de procedure GREEN, wat toegelaten is, maar, vanuit YELLOW kunnen variabelen van GREEN of  PURPLE niet aangesproken worden, vermits YELLOW niet binnen de blokken GREEN of  PURPLE gedeclareerd is. De variabelen van GREEN en PURPLE blijven wel bestaan tijdens de uitvoering van YELLOW, maar zijn niet toegankelijk. Op dat ogenblik zijn enkel de variabelen van YELLOW en van BLUE toegankelijk. Naast de dynamische banden is het dus nodig van een tweede stel adressen bij te houden in de activatie records, deze die, vertrekkend van het lokale activatie record, toelaten de globale variabelen terug te vinden. Daar deze banden het gevolg zijn van de struktuur van het programma en niet van de opeenvolging van de oproepen van procedures tijdens de uitvoering van het programma, noemt men ze de “statische banden”. Figuur 7.11 toont de statische en de dynamische banden die overeenstemmen met de momentopname 6 van figuur 7.10. 

Het is vanzelfsprekend dat de Modula 2 programmeur zich zelf niet hoeft bezig te houden met het opmaken en aanpassen van deze banden, het is de Modula 2 compiler die zorgt voor de code die daarvoor nodig is bij het compileren van procedures en van procedure oproepen.

Fig.7.11. Statische en dynamische banden in het gegevensgeheugen.

7.4. Procedure parameters.

In het voorbeeld werden alle gegevens uitgewisseld tussen het hoofdprogramma en de procedures via globale variabelen. Deze techniek is aanvaardbaar voor kleine programma's, geschreven door één enkel persoon, maar in het algemeen vermindert dit opmerkelijk het nut van procedures, aangezien de auteurs van de procedures en de gebruikers van deze procedures het onderling eens moeten zijn over de namen van de identifiers die ze gaan gebruiken. Bovendien moet de auteur van ieder stuk op de hoogte zijn van wat de anderen gaan doen met de gegevens. Bv., indien men i.p.v. enkel 'The GCD is:' uit te schrijven, men eerder 'The GCD of ... and ... is' verkiest, moeten de initiële waarde van x en y nog steeds beschikbaar zijn na de berekening van de grootste gemene deler, wat niet het geval is aangezien de procedure GCD de waarden van x en y verandert.

7.4.1. Formele en actuele parameters.

Het gebruik van procedure parameters om gegevens uit te wisselen tussen hoofdprogramma en procedures is veiliger en gebruikersvriendelijker dan globale variabelen.

Ten einde de onderliggende principes van procedure parameters beter te begrijpen, kan men zich de richtlijnen indenken om een vriend uit te nodigen voor een avondmaal. In de richtlijnen zal, telkens wanneer de vriend vernoemd wordt, de naam 'YourFriend' gebruikt worden, aangezien de richtlijnen van algemene aard zijn en dat de identiteit van de vriend nog niet gekend is.

De identifier 'YourFriend' noemt men een formele parameter. Van zodra men deze richtlijnen wenst toe te passen, is de naam van de vriend bekend en kan de formele parameter 'YourFriend' vervangen worden door de actuele naam van de vriend die zal uitgenodigd worden. Deze naam noemt men actuele parameter. De implementatie van dit principe wordt getoond in fig.7.12, waar een formele parameter xyz toegevoegd werd aan de procedure DoCCC uit fig.7.1. In het programma zonder procedures vindt men twee maal de code CCC, een maal met de variabelen abc en een maal met de variabelen uvw. In de versie met procedures schrijft men de code CCC met de variabelen xyz en vermeldt men xyz als formele parameter in de procedure hoofding. De actuele parameters (abc voor de eerste activatie, uvw voor de tweede) worden aan de instructies DoCCC toegevoegd, die de procedure activeren. 

Op machine niveau worden de actuele parameters in het gegevensgeheugen geplaatst net na het activatie record van de blok die de oproep bevat. Op die manier bevinden zij zich net voor het activatie record van de opgeroepen procedure en kunnen zij bereikt worden door een negatief adres bij de inhoud van het basisregister te tellen (zie fig.7.13) 

Fig.7.12. PROCEDURE parameters.

Fig.7.13. Het doorgeven van procedure parameters.

7.4.2. Parameter waarvan de waarde wordt doorgegeven (by value).

In het geval enkel waarden aan een procedure moeten doorgegeven worden, bv. de getallen waarvan men de GGD wenst te berekenen, wordt het waarde mechanisme toegepast om de actuele parameters te verbinden met de formele parameters. Met dit mechanisme wordt de formele parameter beschouwd als een lokale variabele van de procedure, waaraan de waarde van de actuele parameter wordt toegekend op het ogenblik dat de procedure wordt aangeroepen. Deze actuele parameter kan een willekeurige uitdrukking zijn die aanleiding geeft tot een waarde die behoort tot het type van de formele parameter.

Een waardetoekenning aan een formele parameter binnen de procedure heeft geen effekt op de actuele parameter vermits, na het beëindigen van de procedure de geheugenruimte waarin de formele parameter bewaard werd gerecycleerd wordt voor het opslaan van andere variabelen.

Het voorbeeld in tabel 7.2 illustreert het doorgeven van de waarde van een parameter.

Een stapsgewijze uitvoering van dit programma wordt beschreven in tabel 7.3 waarin het ??? teken voor de waarde van c wanneer P niet actief is betekent dat c niet bestaat op dat ogenblik.

MODULE TestValPar (* Parameter passing by Value *);
    FROM InOut IMPORT Write,WriteLn;
    VAR x: CHAR;
    PROCEDURE P (c:CHAR);
    BEGIN
        Write(c);
        c:='B';
        Write(c)
        RETURN
    END P;
BEGIN
    x:='A';
    P(x);
    Write(x);
    WriteLn
END TestValPar.

Tabel 7.2. Het doorgeven van de waarde van een parameter.

Line
x
c
Status P
output

12
'A'
???
inactive


13
'A'
'A'
activated


6
'A'
'A'
active
A

7
'A'
'B'
active


8
'A'
'B'
active
B

9
'A'
'B'
deactivated


14
'A'
???
inactive
A

15
'A'
???
inactive
next line

Tabel.7.3. Stapsgewijze uitvoering.

7.4.3. Parameter die als variabele wordt doorgegeven.

Wanneer waarden die in een procedure worden berekend ook buiten die procedure moeten gebruikt worden, kan het mechanisme van het doorgeven van de waarde niet toegepast worden. In dat geval kunnen de parameters doorgegeven worden als variabelen. In dit mechanisme is de formele parameter geen lokale variabele meer binnen de procedure, maar een soort wijzer, waaraan het adres van de actuele parameter werd toegekend op het moment van de aanroep van de procedure. Voor de gebruiker is het alsof, op het ogenblik dat de procedure geactiveerd wordt, overal in de procedure de naam van de formele parameter vervangen werd door de naam van de actuele parameter. Het is evident dat de actuele parameter een variabele van hetzelfde type als de overeenkomstige formele parameter moet zijn.

Aangezien in feite de formele en de actuele parameters enkel maar andere namen zijn voor een zelfde variabele, kan elke waarde toekenning aan een formele parameter binnen de procedure de waarde van de actuele parameter veranderen.

Hetzelfde voorbeeld als in paragraaf  7.4.2, maar met een parameter die als variabele wordt doorgegeven, wordt weergegeven in table 7.4 en de stapsgewijze uitvoering van dat programma, weergegeven in tabel 7.5 toont duidelijk het verschil.

MODULE TestVarPar (* Parameter passing by Variable *);
    FROM InOut IMPORT Write,WriteLn;
    VAR x: CHAR;
    PROCEDURE P (VAR c:CHAR);
    BEGIN
        Write(c);
        c:='B';
        Write(c)
        RETURN
    END P;
BEGIN
    x:='A';
    P(x);
    Write(x);
    WriteLn
END TestVarPar.

Tabel 7.4. Het doorgeven van een parameter als variabele.

Line
x
c
Status P
output

12
'A'
???
inactive


13
'A'
'A'
activated


6
'A'
'A'
active
A

7
'B'
'B'
active


8
'B'
'B'
active
B

9
'B'
'B'
deactivated


14
'B'
???
inactive
B

15
'B'
???
inactive
next line

Tabel.7.5. Stapsgewijze uitvoering.

7.4.4. Nog een andere versie van het GGD programma.

Het is vrij eenvoudig om het GGD programma van paragraaf  7.2.3 aldus te wijzigen zodanig dat de procedure ComputeGCD ook procedure parameters gebruikt. Het aangepaste programma vindt men in tabel 7.6.

Merk op dat het mechanisme van het doorgeven van de waarde gebruikt wordt voor het doorgeven van de waarden van x en y en het mechanisme van het doorgeven van de variabele om de berekende waarde van de waarde van de GCD terug te krijgen.

MODULE GCD;
(* Version with procedures and parameters *)
   FROM InOut IMPORT WriteLn,WriteString,ReadCard,WriteCard;
   VAR x,y,GCDxy:CARDINAL;
   PROCEDURE AskNumbers;
   BEGIN
      WriteString('Enter x and y ');
      ReadCard(x);
      ReadCard(y);
   END AskNumbers;
   PROCEDURE ComputeGCD (a,b:CARDINAL;VAR G:CARDINAL);
   BEGIN
      WHILE a # b DO
         IF a > b 
            THEN a := a-b
            ELSE b := b-a
         END;
      END;
      G := a
   END ComputeGCD;
   PROCEDURE WriteResult;
   BEGIN
      WriteLn;
      WriteString('The GCD of ');
      WriteCard(x,6);
      WriteString(' and ');
      WriteCard(y,6);
      WriteString(' is ');
      WriteCard(GCDxy,6);
   END WriteResult;
BEGIN
   (* Here starts execution of the program *)
   AskNumbers;
   ComputeGCD(x,y,GCDxy);
   WriteResult
END GCD.

Tabel.7.6. Het GGD programma met procedures en parameters.


7.4.5. ARRAYs als procedure parameters.

Gestructureerde variablen kunnen uiteraard ook als parameters aan procedures doorgegeven worden, maar met ARRAYs kan er zich echter een klein praktisch probleempje stellen: wanneer een parameter "by value" doorgegeven wordt, wordt de actuele parameter gecopieerd in de formele, wat voor grote ARRAYs veel plaats en tijd kan kosten. Het is dus aan te bevelen, zelfs wanneer de actuele parameter niet gewijzigd wordt door de procedure, toch het VAR mechanisme te gebruiken voor grote ARRAYs. Er dient dan wel op gelet te worden dat de ARRAY niet in de procedure gewijzigd wordt.

7.4.6. Open ARRAYs.

De strikte identiteit van types tussen actuele en formele parameters is, in het geval van één dimensionale ARRAYs die met het VAR mechanisme doorgegeven worden, overdreven streng en kan tot absurde programma's leiden: veronderstel dat men in een programma twee ARRAYs, één met twintig getallen en één met dertig getallen, wenst te ordenen. Een procedure hiervoor schrijven en de ARRAYs als parameters doorgeven lijkt aangewezen, maar vermits actuele en formele parameters strikt van hetzelfde type moeten zijn, zouden er twee procedures nodig zijn, één voor de ARRAY met twintig elementen en één voor die met dertig. Deze twee procedures zouden enkel maar van elkander verschillen door het type van hun formele parameter en door een constante die de afmeting van de ARRAY aangeeft. Op het niveau van de object code zou er zelfs nog minder verschil zijn, vermits, bij het gebruik van het VAR mechanisme, de formele parameter alleen maar het adres van het eerste element van de actuele parameter bevat, onafgezien van de omvang van deze laatste.

Modula 2 biedt een elegante oplossing voor deze laatste moeilijkheid: het is toegelaten een één dimensionale ARRAY als formele VAR parameter te declareren zonder het type van de index te vermelden. Gelijk welke één dimensionale ARRAY met elementen van hetzelfde type kan dan als actuele parameter gebruikt worden zonder dat er een type fout zou optreden.

De hoofding van een procedure die gelijk welke ARRAY OF INTEGERs kan ordenen zou er als volgt uitzien:

PROCEDURE Sort (VAR A:ARRAY OF INTEGER);

Wanneer men in een procedure een "open ARRAY" gebruikt is het erg nuttig ook de omvang van de actuele parameter te kennen. Dit kan door gebruik te maken van de functie-procedure HIGH. Deze CARDINAL functie, waarvan de parameter de naam van een "open ARRAY" moet zijn, krijgt de waarde van de hoogste index van de actuele parameter, in de veronderstelling dat de laagste de waarde 0 heeft.

In de procedure Sort die als voorbeeld hierboven vermeld werd zou de functie HIGH(A) de waarden 20, 30, 19 en 6 krijgen wanneer deze procedure resp. met de actuele parameters X,Y,Z en W uit tabel 7.7 zouden opgeroepen worden, waarbij het type Day een enumeratie van de zeven dagen van de week is.

X : ARRAY[0..20] OF INTEGER;

Y : ARRAY[0..30] OF INTEGER;

Z : ARRAY[1..20] OF INTEGER;

W : ARRAY[Day] OF INTEGER;

Tabel 7.7. Voorbeelden van actuele parameters
7.5. Functies.

In het voorbeeld in paragraaf 7.4.4 kan men vaststellen dat het terugkrijgen van de waarde van de GGD via een variabele parameter nogal onpraktisch is: zowel de actuele als de formele parameter werden gedeclareerd enkel en alleen om de waarde van de GGD buiten de procedure te kennen.

7.5.1. Functie procedures.

Wanneer een procedure slechts een enkel resultaat moet geven, kan men dit doen via de RETURN instructie, eerder dan via een variabele parameter. In dit geval wordt de procedure dikwijls een functie ('function') genoemd, alhoewel deze naam niet voorkomt in de syntax van Modula 2.

Een functie heeft een type, dat de verzameling aangeeft van alle mogelijke waarden die de functie kan aannemen, en moet een RETURN instructie bevatten, gevolgd door een uitdrukking, waarvan de waarde behoort tot het type van de functie (fig.7.14).

Fig.7.14. Bijzonderheden van de functie syntax.[image: image14.wmf]Function PROCEDURE heading.

RETURN statement.


Functies kunnen voorkomen in uitdrukkingen en worden geactiveerd op het ogenblik dat de uitdrukking geevalueerd wordt.

7.5.2. Een laatste versie van het GGD programma, door gebruik te maken van functies.

MODULE GCD;
(* Version with procedures, functions and parameters *)
   FROM InOut IMPORT WriteLn, WriteString, ReadCard, WriteCard;
   VAR x,y,:CARDINAL;
   PROCEDURE AskNumber: CARDINAL;
      VAR x : CARDINAL;
   BEGIN
      WriteString('Enter a number ');
      ReadCard(x);
      RETURN x;
   END AskNumber;
   PROCEDURE ComputeGCD (a,b:CARDINAL):CARDINAL;
   BEGIN
      WHILE a # b DO
         IF a > b 
            THEN a := a-b
            ELSE b := b-a
         END;
      END;
      RETURN a
   END ComputeGCD;
   PROCEDURE WriteResult;
   BEGIN
      WriteLn;
      WriteString('The GCD of ');
      WriteCard(x,6);
      WriteString(' and ');
      WriteCard(y,6);
      WriteString(' is ');
      WriteCard(ComputeGCD(x,y),6);
   END WriteResult;
BEGIN
   (* Here starts execution of the program *)
   x := AskNumber; y := AskNumber;
   WriteResult
END GCD.

Fig.7.19. Het GGD programma met procedures en parameters.

Merk op dat, als men optimaal gebruik wenst te maken van de eigenschappen van functies, de structuur van het programma aangepast moet worden: de procedure AskNumber vraagt nog maar een getal op en wordt dus twee maal aangeroepen terwijl de procedure ComputeGCD niet meer wordt aangeroepen vanuit het hoofdprogramma, maar vanuit de WriteResult procedure, als parameter van de WriteCard procedure.

7.5.3. Neven verschijnselen (Side effects).

Wanneer men in een programma een functie procedure ontmoet zoals SIN(x), LOG(x) of ComputeGCD(x,y), verwacht men intuitief dat deze procedure een zekere functie zal evalueren en niets anders. Men verwacht geenszins dat de parameters van de functie, of erger, sommige globale variabelen kunnen gewijzigd worden tijdens de evaluatie van de functie. Desalniettemin, aangezien functie procedures normale procedures zijn, die VAR parameters kunnen hebben, en waarop de normale draagwijdte regels voor identifiers van toepassing zijn, kunnen ze veel meer doen dan enkel maar een waarde teruggeven. Deze bijkomende verschijnselen van een functie procedure worden 'side-effects' genoemd. In traditionele imperatieve gestructureerde programmering, moeten side-effects zo veel mogelijk vermeden worden, aangezien ze de neiging hebben de programma's moeilijker te verstaan te maken.

